

F07MGF (SSYCON/DSYCON) – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

F07MGF (SSYCON/DSYCON) estimates the condition number of a real symmetric indefinite matrix A , where A has been factorized by F07MDF (SSYTRF/DSYTRF).

2 Specification

```

SUBROUTINE F07MGF(UPLO, N, A, LDA, IPIV, ANORM, RCOND, WORK,
1              IWORK, INFO)
ENTRY      ssycon(UPLO, N, A, LDA, IPIV, ANORM, RCOND, WORK,
1              IWORK, INFO)
INTEGER    N, LDA, IPIV(*), IWORK(*), INFO
real      A(LDA,*), ANORM, RCOND, WORK(*)
CHARACTER*1 UPLO

```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine estimates the condition number (in the 1-norm) of a real symmetric indefinite matrix A :

$$\kappa_1(A) = \|A\|_1 \|A^{-1}\|_1.$$

Since A is symmetric, $\kappa_1(A) = \kappa_\infty(A) = \|A\|_\infty \|A^{-1}\|_\infty$.

Because $\kappa_1(A)$ is infinite if A is singular, the routine actually returns an estimate of the **reciprocal** of $\kappa_1(A)$.

The routine should be preceded by a call to F06RCF to compute $\|A\|_1$ and a call to F07MDF (SSYTRF/DSYTRF) to compute the Bunch–Kaufman factorization of A . The routine then uses Higham's implementation of Hager's method [1] to estimate $\|A^{-1}\|_1$.

4 References

- [1] Higham N J (1988) FORTRAN codes for estimating the one-norm of a real or complex matrix, with applications to condition estimation *ACM Trans. Math. Software* **14** 381–396

5 Parameters

1: UPLO — CHARACTER*1 *Input*

On entry: indicates how A has been factorized as follows:

if UPLO = 'U', then $A = PUDU^T P^T$, where U is upper triangular;

if UPLO = 'L', then $A = PLDL^T P^T$, where L is lower triangular.

Constraint: UPLO = 'U' or 'L'.

2: N — INTEGER *Input*

On entry: n , the order of the matrix A .

Constraint: $N \geq 0$.

- 3:** A(LDA,*) — *real* array *Input*
Note: the second dimension of the array A must be at least $\max(1,N)$.
On entry: details of the factorization of A , as returned by F07MDF (SSYTRF/DSYTRF).
- 4:** LDA — INTEGER *Input*
On entry: the first dimension of the array A as declared in the (sub)program from which F07MGF (SSYCON/DSYCON) is called.
Constraint: $LDA \geq \max(1,N)$.
- 5:** IPIV(*) — INTEGER array *Input*
Note: the dimension of the array IPIV must be at least $\max(1,N)$.
On entry: details of the interchanges and the block structure of D , as returned by F07MDF (SSYTRF/DSYTRF).
- 6:** ANORM — *real* *Input*
On entry: the 1-norm of the **original** matrix A , which may be computed by calling F06RCF. ANORM must be computed either **before** calling F07MDF (SSYTRF/DSYTRF) or else from a copy of the original matrix A .
Constraint: $ANORM \geq 0.0$.
- 7:** RCOND — *real* *Output*
On exit: an estimate of the reciprocal of the condition number of A . RCOND is set to zero if exact singularity is detected or the estimate underflows. If RCOND is less than **machine precision**, then A is singular to working precision.
- 8:** WORK(*) — *real* array *Workspace*
Note: the dimension of the array WORK must be at least $\max(1,2*N)$.
- 9:** IWORK(*) — INTEGER array *Workspace*
Note: the dimension of the array IWORK must be at least $\max(1,N)$.
- 10:** INFO — INTEGER *Output*
On exit: $INFO = 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If $INFO = -i$, the i th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed estimate RCOND is never less than the true value ρ , and in practice is nearly always less than 10ρ , although examples can be constructed where RCOND is much larger.

8 Further Comments

A call to this routine involves solving a number of systems of linear equations of the form $Ax = b$; the number is usually 4 or 5 and never more than 11. Each solution involves approximately $2n^2$ floating-point operations but takes considerably longer than a call to F07MEF (SSYTRS/DSYTRS) with 1 right-hand side, because extra care is taken to avoid overflow when A is approximately singular.

The complex analogues of this routine are F07MUF (CHECON/ZHECON) for Hermitian matrices and F07NUF (CSYCON/ZSYCON) for symmetric matrices.

9 Example

To estimate the condition number in the 1-norm (or infinity-norm) of the matrix A , where

$$A = \begin{pmatrix} 2.07 & 3.87 & 4.20 & -1.15 \\ 3.87 & -0.21 & 1.87 & 0.63 \\ 4.20 & 1.87 & 1.15 & 2.06 \\ -1.15 & 0.63 & 2.06 & -1.81 \end{pmatrix}.$$

Here A is symmetric indefinite and must first be factorized by F07MDF (SSYTRF/DSYTRF). The true condition number in the 1-norm is 75.68.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      F07MGF Example Program Text
*      Mark 15 Release. NAG Copyright 1991.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER       (NIN=5,NOUT=6)
      INTEGER          NMAX, LDA, LWORK
      PARAMETER       (NMAX=8,LDA=NMAX,LWORK=64*NMAX)
*      .. Local Scalars ..
      real            ANORM, RCOND
      INTEGER          I, INFO, J, N
      CHARACTER        UPLO
*      .. Local Arrays ..
      real            A(LDA,NMAX), WORK(LWORK)
      INTEGER          IPIV(NMAX), IWORK(NMAX)
*      .. External Functions ..
      real            F06RCF, X02AJF
      EXTERNAL         F06RCF, X02AJF
*      .. External Subroutines ..
      EXTERNAL         ssycon, ssytrf
*      .. Executable Statements ..
      WRITE (NOUT,*) 'F07MGF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*          Read A from data file
*
*          READ (NIN,*) UPLO
*          IF (UPLO.EQ.'U') THEN
*              READ (NIN,*) ((A(I,J),J=I,N),I=1,N)
*          ELSE IF (UPLO.EQ.'L') THEN
*              READ (NIN,*) ((A(I,J),J=1,I),I=1,N)
*          END IF
*
*          Compute norm of A
*
*          ANORM = F06RCF('1-norm',UPLO,N,A,LDA,WORK)
*

```

```

*      Factorize A
*
      CALL ssytrf(UPLO,N,A,LDA,IPIV,WORK,LWORK,INFO)
*
      WRITE (NOUT,*)
      IF (INFO.EQ.0) THEN
*
*      Estimate condition number
*
      CALL ssycon(UPLO,N,A,LDA,IPIV,ANORM,RCOND,WORK,IWORK,INFO)
*
      IF (RCOND.GE.X02AJF()) THEN
          WRITE (NOUT,99999) 'Estimate of condition number =',
+          1.0e0/RCOND
      ELSE
          WRITE (NOUT,*) 'A is singular to working precision'
      END IF
      ELSE
          WRITE (NOUT,*) 'The factor D is singular'
      END IF
      END IF
      STOP
*
99999 FORMAT (1X,A,1P,e10.2)
      END

```

9.2 Program Data

```

F07MGF Example Program Data
4                               :Value of N
'L'                             :Value of UPLO
2.07
3.87 -0.21
4.20  1.87  1.15
-1.15  0.63  2.06 -1.81      :End of matrix A

```

9.3 Program Results

```

F07MGF Example Program Results

Estimate of condition number = 7.57E+01

```
